In[1]:= `NotebookDirectory[]`

Out[1]= `C:\Dropbox\Work\myweb\Courses\Math_pages\Math_430\`

Equilibrium temperature distribution - 2D problem

The problem

The objective is to solve the PDE

$$\frac{\partial^2 u}{\partial x^2}(x, y) + \frac{\partial^2 u}{\partial y^2}(x, y) = 0 \text{ on } \{(x, y) \in \ \mid 0 \leq x \leq K,\ 0 \leq y \leq L\},$$

subject to the conditions

$u(x, 0) = f_1(x),\ u(x, L) = f_2(x)$ (call these **BCx**)

$u(0, y) = g_1(y),\ u(K, y) = g_2(y)$ (call these **BCy**)

The trick is to split this problem into **two problems**

Problem 1

The objective is to solve the PDE

$$\frac{\partial^2 u}{\partial x^2} + \frac{\partial^2 u}{\partial y^2} = 0 \text{ on } 0 \leq x \leq K,\ 0 \leq y \leq L,$$

subject to the conditions

$u(x, 0) = f_1(x),\ u(x, L) = f_2(x)$ (call these **BCx**)

$u(0, y) = 0,\ u(K, y) = 0$ (call these **DBCy**)

Step 1. First ignore **BCx** and solve

$$\frac{\partial^2 u}{\partial x^2} + \frac{\partial^2 u}{\partial y^2} = 0 \text{ on } 0 \le x \le K,\ 0 \le y \le L,$$

subject to the conditions

$u(0, y) = 0,\ u(K, y) = 0$ (call these **DBCy**)

Using the Separation of Variables (SofV) method we find few solutions of this problem:

$$\text{Sin}\left[\frac{n\,\text{Pi}}{K}\,x\right]\frac{\text{Sinh}\left[\frac{n\,\text{Pi}}{K}\,y\right]}{\text{Sinh}\left[\frac{n\,\text{Pi}}{K}\,L\right]}$$

$$\text{Sin}\left[\frac{n\,\text{Pi}}{K}\,x\right]\frac{\text{Sinh}\left[\frac{n\,\text{Pi}}{K}\,(L-y)\right]}{\text{Sinh}\left[\frac{n\,\text{Pi}}{K}\,L\right]}$$

Test these solutions:

```
In[2]:= (D[#, {x, 2}] + D[#, {y, 2}]) &[Sin[n Pi/K x] Sinh[n Pi/K y]/Sinh[n Pi/K L]]

Out[2]= 0
```

In[3]:=
```
FullSimplify[(Sin[n Pi/K x] Sinh[n Pi/K y]/Sinh[n Pi/K L]) /. {x → {0, K}}]
```

Out[3]= $\left\{0, \operatorname{Csch}\left[\frac{L n \pi}{K}\right] \operatorname{Sin}[n \pi] \operatorname{Sinh}\left[\frac{n \pi y}{K}\right]\right\}$

We need to tell *Mathematica* that n is an integer.

In[4]:=
```
FullSimplify[(Sin[n Pi/K x] Sinh[n Pi/K y]/Sinh[n Pi/K L]) /. {x → {0, K}},
  n ∈ Integers]
```

Out[4]= {0, 0}

In[5]:=
```
(D[#, {x, 2}] + D[#, {y, 2}]) &[Sin[n Pi/K x] Sinh[n Pi/K (L - y)]/Sinh[n Pi/K L]]
```

Out[5]= 0

In[6]:=
```
FullSimplify[(Sin[n Pi/K x] Sinh[n Pi/K (L - y)]/Sinh[n Pi/K L]) /. {x → {0, K}},
  n ∈ Integers]
```

Out[6]= {0, 0}

Step 2. Now that we have few solutions we form many solutions. This is the From-Few-Many (FFM) idea, which is commonly known as the superposition principle:

$$\sum_{n=1}^{nn} a_n \operatorname{Sin}\left[\frac{n\,Pi}{K} x\right] \frac{\operatorname{Sinh}\left[\frac{n\,Pi}{K} y\right]}{\operatorname{Sinh}\left[\frac{n\,Pi}{K} L\right]} +$$

$$\sum_{n=1}^{nn} b_n \text{Sin}\left[\frac{n\,Pi}{K}\,x\right] \frac{\text{Sinh}\left[\frac{n\,Pi}{K}\,(L-y)\right]}{\text{Sinh}\left[\frac{n\,Pi}{K}\,L\right]}$$

Next we choose a_n and b_n such that the above function satisfies **BCx** conditions. First substitute $y = 0$. This leads to the formula for b_n.

$$f_1\,(x) \;=\; \sum_{n=1}^{nn} b_n \text{Sin}\left[\frac{n\,Pi}{K}\,x\right]$$

$$f_1\,(x)\,\text{Sin}\left[\frac{j\,Pi}{K}\,x\right] \;=\; \sum_{n=1}^{nn} b_n \text{Sin}\left[\frac{n\,Pi}{K}\,x\right] \text{Sin}\left[\frac{j\,Pi}{K}\,x\right]$$

In[7]:=
```
Clear[K];
FullSimplify[Integrate[Sin[n Pi/K x] Sin[j Pi/K x], {x, 0, K}],
 And[n ∈ Integers, j ∈ Integers, Or[j > n, j < n]]]
```

Out[7]= 0

In[8]:=
```
Clear[K];
FullSimplify[Integrate[Sin[n Pi/K x] Sin[n Pi/K x], {x, 0, K}],
 And[n ∈ Integers]]
```

Out[8]= $\frac{K}{2}$

$$\int_0^K f_1\,(x)\,\text{Sin}\left[\frac{j\,Pi}{K}\,x\right] dx \;=$$

$$\sum_{n=1}^{nn} b_n \int_0^K \text{Sin}\left[\frac{n\,Pi}{K}\,x\right] \text{Sin}\left[\frac{j\,Pi}{K}\,x\right] dx$$

$$\int_0^K f_1\,(x)\,\text{Sin}\left[\frac{j\,Pi}{K}\,x\right] dx \;=\; b_n \int_0^K \text{Sin}\left[\frac{j\,Pi}{K}\,x\right] \text{Sin}\left[\frac{j\,Pi}{K}\,x\right] dx$$

$$\int_0^K f_1\,(x)\,\text{Sin}\left[\frac{j\,Pi}{K}\,x\right] dx \;=\; b_j * \frac{K}{2}$$

Then substitute $y = L$. This leads to the formula for a_n.

Problem 2

The objective is to solve the PDE

$$\frac{\partial^2 u}{\partial x^2} + \frac{\partial^2 u}{\partial y^2} = 0 \text{ on } 0 \le x \le K, \ 0 \le y \le L,$$

subject to the conditions

$u(x, 0) = 0, \ u(x, L) = 0$ (call these **DBCx**)

$u(0, y) = g_1(y), \ u(K, y) = g_2(y)$ (call these **BCy**)

Step 1. First ignore **BCy** and solve

$$\frac{\partial^2 u}{\partial x^2} + \frac{\partial^2 u}{\partial y^2} = 0 \text{ on } 0 \le x \le K, \ 0 \le y \le L,$$

subject to the conditions

$u(x, 0) = 0, \ u(x, L) = 0$ (call these **DBCx**)

Using SofV method we find few solutions of this problem:

$$\mathrm{Sin}\left[\frac{n\,\mathrm{Pi}}{L}\,y\right]\,\frac{\mathrm{Sinh}\left[\frac{n\,\mathrm{Pi}}{L}\,x\right]}{\mathrm{Sinh}\left[\frac{n\,\mathrm{Pi}}{L}\,K\right]}$$

$$\mathrm{Sin}\left[\frac{n\,\mathrm{Pi}}{L}\,y\right]\,\frac{\mathrm{Sinh}\left[\frac{n\,\mathrm{Pi}}{L}\,(K-x)\right]}{\mathrm{Sinh}\left[\frac{n\,\mathrm{Pi}}{L}\,K\right]}$$

Test these solutions:

```
In[9]:= (D[#, {x, 2}] + D[#, {y, 2}]) &[Sin[n Pi/L y] Sinh[n Pi/L x]/Sinh[n Pi/L K]]

Out[9]= 0

In[10]:= FullSimplify[(Sin[n Pi/L y] Sinh[n Pi/L x]/Sinh[n Pi/L K]) /. {y → {0, L}},
   n ∈ Integers]

Out[10]= {0, 0}

In[11]:= (D[#, {x, 2}] + D[#, {y, 2}]) &[Sin[n Pi/L y] Sinh[n Pi/L (K - x)]/Sinh[n Pi/L K]]

Out[11]= 0

In[12]:= FullSimplify[(Sin[n Pi/L y] Sinh[n Pi/L (K - x)]/Sinh[n Pi/L K]) /. {y → {0, L}},
   n ∈ Integers]

Out[12]= {0, 0}
```

Step 2. Now that we have few solutions we form many solutions. This is the FFM idea, which is commonly known as the superposition principle:

$$\sum_{n=1}^{nn} c_n \operatorname{Sin}\left[\frac{n\,Pi}{L} y\right] \frac{\operatorname{Sinh}\left[\frac{n\,Pi}{L} x\right]}{\operatorname{Sinh}\left[\frac{n\,Pi}{L} K\right]} + \sum_{n=1}^{nn} d_n \operatorname{Sin}\left[\frac{n\,Pi}{L} y\right] \frac{\operatorname{Sinh}\left[\frac{n\,Pi}{L} (K - x)\right]}{\operatorname{Sinh}\left[\frac{n\,Pi}{L} K\right]}$$

Now we choose c_n and d_n such that the above function satisfies **BCy** conditions. First substitute $x = 0$. This leads to the formula for d_n. Then substitute $x = K$. This leads to the formula for c_n.

A symbolic implementation

Here are the given quantities

In[13]:=
```
Clear[lK1, lL1, f11, f21, g11, g21, nn1];

nn1 = 15;

lK1 = 1; lL1 = 1;

f11[x_] = 4 x^2 (1 - x);

g21[y_] = 4 y (1 - y)^2;

f21[x_] = 0;

g11[y_] = 0;
```

In[20]:=
```
Clear[aa1];

aa1[n_] =
 FullSimplify[
  2/lK1 Integrate[f21[x] Sin[n Pi/lK1 x], {x, 0, lK1}],
  And[n ∈ Integers, n > 0]]
```

Out[21]= 0

In[22]:=
```
Clear[bb1];

bb1[n_] =
 FullSimplify[
  2/lK1 Integrate[f11[x] Sin[n Pi/lK1 x], {x, 0, lK1}],
  And[n ∈ Integers, n > 0]]
```

Out[23]= $-\frac{16\left(1+2(-1)^{n}\right)}{n^{3}\pi^{3}}$

In[24]:=
```
Clear[cc1];

cc1[n_] =
 FullSimplify[
  2/lL1 Integrate[g21[y] Sin[n Pi/lL1 y], {y, 0, lL1}],
  And[n ∈ Integers, n > 0]]
```

Out[25]= $\frac{16\left(2+(-1)^{n}\right)}{n^{3}\pi^{3}}$

In[26]:=
```
Clear[dd1];

dd1[n_] =
 FullSimplify[
  2/lL1 Integrate[g11[y] Sin[n Pi/lL1 y], {y, 0, lL1}],
  And[n ∈ Integers, n > 0]]
```

Out[27]= 0

The solution is

In[28]:=
```
Clear[uu1];
```

```
uu1[x_, y_] = Sum[aa1[n] Sin[n Pi/lK1 x] Sinh[n Pi/lK1 y]/Sinh[n Pi/lK1 lL1], {n, 1, nn1}] +
   Sum[bb1[n] Sin[n Pi/lK1 x] Sinh[n Pi/lK1 (lL1 - y)]/Sinh[n Pi/lK1 lL1], {n, 1, nn1}] +
   Sum[cc1[n] Sin[n Pi/lL1 y] Sinh[n Pi/lL1 x]/Sinh[n Pi/lL1 lK1], {n, 1, nn1}] +
   Sum[dd1[n] Sin[n Pi/lL1 y] Sinh[n Pi/lL1 (lK1 - x)]/Sinh[n Pi/lL1 lK1], {n, 1, nn1}];
```

How good is our approximation for the function f1[x] in the boundary conditions? Here is a visual answer.

In[30]:=
```
Plot[{f11[x], uu1[x, 0]}, {x, 0, 1},
 PlotStyle → {{Blue, Thickness[0.01]},
   {Red, Thickness[0.005]}}]
```

Out[30]=



In[31]:= `Plot[uu1[x, lL], {x, 0, 1}]`

Out[31]=



In[32]:= `Plot[uu1[0, x], {x, 0, 1}]`

Out[32]=



In[33]:=
```
Plot[{g21[y], uu1[lK, y]}, {y, 0, 1},
  PlotStyle → {{Blue, Thickness[0.01]},
    {Red, Thickness[0.005]}}]
```

Out[33]=



In[34]:=
```
Plot3D[N[uu1[x, y]], {x, 0, 1}, {y, 0, 1}, Mesh → Automatic]
```

Out[34]=



In[35]:=
```
DensityPlot[N[uu1[x, y]], {x, 0, 1}, {y, 0, 1},
 Frame → True, PlotRange → {{0, 1}, {0, 1}},
 ColorFunction → (RGBColor[1, 1 - #, 1 - #] &)]
```

Out[35]=



A numerical implementation

Here are the given quantities

In[36]:=
```
Clear[lK2, lL2, f12, f22, g12, g22, nn2];

nn2 = 45;

lK2 = Pi; lL2 = Pi;

f12[x_] = (Sin[x])^(1/10);
g22[y_] = 0; f22[x_] = 0;
g12[y_] = 0;
```

In[40]:=
```
Plot[f12[x], {x, 0, lK2}, PlotRange → {0, 1.2}]
```

Out[40]=



In[41]:=
```
Clear[aa2];
```

$$aa2[n\_] = \frac{2}{lK2}\, \text{Integrate}\left[f22[x]\, \text{Sin}\left[\frac{n\, Pi}{lK2}\, x\right], \{x, 0, lK2\}\right]$$

Out[42]= 0

In[43]:=
```
Clear[bbl2];

bbl2 =
 Chop[Table[2/lK2 NIntegrate[f12[x] Sin[n Pi/lK2 x], {x, 0, lK2},
    Method → {Automatic}, MaxRecursion → 200,
    AccuracyGoal → 12, PrecisionGoal → 16], {n, 1, nn2}]]
```

Out[44]=
```
{1.23582, 0, 0.358787, 0, 0.204016, 0, 0.1408, 0,
 0.10676, 0, 0.0856006, 0, 0.0712249, 0, 0.0608478, 0,
 0.0530194, 0, 0.0469124, 0, 0.0420211, 0, 0.0380191,
 0, 0.0346867, 0, 0.0318708, 0, 0.0294614, 0,
 0.0273773, 0, 0.0255576, 0, 0.0239557, 0, 0.0225352, 0,
 0.0212672, 0, 0.0201288, 0, 0.0191014, 0, 0.0181696}
```

In[45]:=
```
Clear[cc2];

cc2[n_] = 2/(lL2 Sinh[n Pi/lL2 lK2])
  Integrate[g22[y] Sin[n Pi/lL2 y], {y, 0, lL2}]
```

Out[46]= 0

In[47]:=
```
Clear[dd2];

dd2[n_] = 2/lL2 Integrate[g12[y] Sin[n Pi/lL2 y], {y, 0, lL2}]
```

Out[48]= 0

The solution is

In[49]:= Clear[uu2];

$$uu2[x\_, y\_] = \sum_{n=1}^{nn2} aa2[n] \, Sin\left[\frac{n \, Pi}{lK2} x\right] \frac{Sinh\left[\frac{n \, Pi}{lK2} y\right]}{Sinh\left[\frac{n \, Pi}{lK2} lL2\right]} + \sum_{n=1}^{nn2} bbl2[[n]] \, Sin\left[\frac{n \, Pi}{lK2} x\right] \frac{Sinh\left[\frac{n \, Pi}{lK2} (lL2 - y)\right]}{Sinh\left[\frac{n \, Pi}{lK2} lL2\right]} + \sum_{n=1}^{nn2} cc2[n] \, Sin\left[\frac{n \, Pi}{lL2} y\right] \frac{Sinh\left[\frac{n \, Pi}{lL2} x\right]}{Sinh\left[\frac{n \, Pi}{lL2} lK2\right]} + \sum_{n=1}^{nn2} dd2[n] \, Sin\left[\frac{n \, Pi}{lL2} y\right] \frac{Sinh\left[\frac{n \, Pi}{lL2} (lK2 - x)\right]}{Sinh\left[\frac{n \, Pi}{lL2} lK2\right]};$$

In[51]:= Plot[{f12[x], uu2[x, 0]}, {x, 0, lK2}, PlotRange → {0, 1.2}]

Out[51]=



In[52]:=
```
Plot[uu2[x, lL2], {x, 0, lK2}]
```

Out[52]=



In[53]:=
```
Plot[uu2[0, x], {x, 0, lK2}]
```

Out[53]=



In[54]:= `Plot[uu2[lK, x], {x, 0, lL2}]`

Out[54]=



In[55]:= `Plot3D[N[uu2[x, y]], {x, 0, lK2}, {y, 0, lL2}]`

Out[55]=



In[56]:=
```
DensityPlot[N[uu2[x, y]], {x, 0, lK2}, {y, 0, lL2},
 Frame → True, PlotRange → {{0, lK2}, {0, lL2}},
 ColorFunction → (RGBColor[1, 1 - #, 1 - #] &)]
```

Out[56]=



A symbolic implementation with a problem

Here are the given quantities

```
In[57]:= Clear[lK3, lL3, f13, f23, g13, g23, nn3];

nn3 = 20;

lK3 = 1; lL3 = 1;

f13[x_] = 3 x^2 + 1;

g23[y_] = 4 - 8 y (y - 1);

f23[x_] = 4 + 8 x (x - 1);

g13[y_] = 1 + 3 y^2;

In[64]:= Clear[aa3];

aa3[n_] =
 FullSimplify[
  2/lK3 Integrate[f23[x] Sin[n Pi/lK3 x], {x, 0, lK3}],
  And[n ∈ Integers, n > 0]]
```

Out[65]= $-\frac{8\left(-1+(-1)^{n}\right)\left(-4+n^{2}\pi^{2}\right)}{n^{3}\pi^{3}}$

In[66]:=
```
Clear[bb3];

bb3[n_] =
 FullSimplify[
  2/lK3 Integrate[f13[x] Sin[n Pi/lK3 x], {x, 0, lK3}],
  And[n ∈ Integers, n > 0]]
```

Out[67]= $$\frac{2\left(6\left(-1+(-1)^{n}\right)+\left(1-4(-1)^{n}\right)n^{2}\pi^{2}\right)}{n^{3}\pi^{3}}$$

In[68]:=
```
Clear[cc3];

cc3[n_] =
 FullSimplify[
  2/lL3 Integrate[g23[y] Sin[n Pi/lL3 y], {y, 0, lL3}],
  And[n ∈ Integers, n > 0]]
```

Out[69]= $$-\frac{8\left(-1+(-1)^{n}\right)\left(4+n^{2}\pi^{2}\right)}{n^{3}\pi^{3}}$$

In[70]:=
```
Clear[dd3];

dd3[n_] = 2/lL3 Integrate[g13[y] Sin[n Pi/lL3 y], {y, 0, lL3}]
```

Out[71]= $$\frac{2\left(-6+n^{2}\pi^{2}+\left(6-4n^{2}\pi^{2}\right)\mathrm{Cos}[n\pi]+6n\pi\,\mathrm{Sin}[n\pi]\right)}{n^{3}\pi^{3}}$$

The solution is

In[72]:=
```
Clear[uu3];
```

$$uu3[x\_, y\_] = \sum_{n=1}^{nn3} aa3[n] \, Sin\left[\frac{n \, Pi}{lK3} x\right] \frac{Sinh\left[\frac{n \, Pi}{lK3} y\right]}{Sinh\left[\frac{n \, Pi}{lK3} lL3\right]} + \sum_{n=1}^{nn3} bb3[n] \, Sin\left[\frac{n \, Pi}{lK3} x\right] \frac{Sinh\left[\frac{n \, Pi}{lK3} (lL3 - y)\right]}{Sinh\left[\frac{n \, Pi}{lK3} lL3\right]} + \sum_{n=1}^{nn3} cc3[n] \, Sin\left[\frac{n \, Pi}{lL3} y\right] \frac{Sinh\left[\frac{n \, Pi}{lL3} x\right]}{Sinh\left[\frac{n \, Pi}{lL3} lK3\right]} + \sum_{n=1}^{nn3} dd3[n] \, Sin\left[\frac{n \, Pi}{lL3} y\right] \frac{Sinh\left[\frac{n \, Pi}{lL3} (lK3 - x)\right]}{Sinh\left[\frac{n \, Pi}{lL3} lK3\right]};$$

How good are the approximations? Here are visual answers:

In[74]:=
```
Plot[{f13[x], uu3[x, 0]}, {x, 0, lK3},
 PlotStyle → {{Blue, Thickness[0.01]},
   {Red, Thickness[0.005]}}, PlotRange → All]
```

Out[74]=



In[75]:=
```
Plot[{g23[y], uu3[lK3, y]}, {y, 0, lL3},
 PlotStyle → {{Blue, Thickness[0.01]},
   {Red, Thickness[0.005]}}, PlotRange → All]
```

Out[75]=



In[76]:=
```
Plot[{f23[x], uu3[x, lL3]}, {x, 0, lK3},
 PlotStyle → {{Blue, Thickness[0.01]},
   {Red, Thickness[0.005]}}, PlotRange → All]
```

Out[76]=



In[77]:=
```
Plot[{g13[y], uu3[0, y]}, {y, 0, lL3},
 PlotStyle → {{Blue, Thickness[0.01]},
   {Red, Thickness[0.005]}}, PlotRange → All]
```

Out[77]=



In[78]:=
```
Plot3D[N[uu3[x, y]], {x, 0, 1K3}, {y, 0, 1L3},
  Mesh → Automatic, PlotRange → {0, 6.5}]
```

Out[78]=



In[79]:=
```
DensityPlot[N[uu3[x, y]], {x, 0, lK3}, {y, 0, lL3},
  Frame → True, PlotRange → {{0, lK3}, {0, lL3}},
  ColorFunction → (RGBColor[1, 1 - #, 1 - #] &)]
```

Out[79]=